

Zählfunktionen in Excel

Inhalt

Mit ZÄHLENWENN() Excel-Listen einfach auswerten	2
Die Excel-Funktion ZÄHLENWENNS()	9
Zellen zählen in Excel – mit ODER-Bedingung	14
Ungerade Werte in einer Excel-Liste zählen	17
Unterschiedliche Werte in einer Excel-Liste zählen	21
Werte mit Excel außerhalb der Toleranz zählen (Toleranzformel)	27
Werte mit Excel außerhalb eines Intervalls zählen	32
Zellen in Excel in Abhängigkeit der Hintergrundfarbe zählen oder summieren	36
Empfehlungen aus dem Management-Handbuch	45

Mit ZÄHLENWENN() Excel-Listen einfach auswerten

Mit der Excel-Funktion ZÄHLENWENN() zählen Sie schnell und einfach Einträge in Listen aus. So erkennen Sie, wie viele Einträge eines bestimmten Werts es gibt. Dabei sind auch Größer- und Kleiner-Vergleiche möglich.

Zuletzt geändert am 18.03.2026



Die Funktion **ZÄHLENWENN()** ist eine der statistischen Funktionen in Excel, mit der Sie die Anzahl von Zellen zählen können, die ein bestimmtes Kriterium erfüllen.

ZÄHLENWENN() ist eine der wichtigsten Funktionen für die tägliche Praxis. Die folgenden Beispiele zeigen, wie die Excel-Funktion eingesetzt wird und was sie leistet.

Aufbau und Funktionsweise von ZÄHLENWENN()

Die Syntax von ZÄHLENWENN() lautet:

=ZÄHLENWENN(Bereich; Suchkriterien)


- **Bereich:** Wo möchten Sie suchen? Hier erfassen Sie den Bereich, der nach einem bestimmten Kriterium durchsucht werden soll.
- **Suchkriterien:** Wonach möchten Sie suchen? Erfassen Sie hier das Kriterium, das im Bereich gesucht und gezählt werden soll.

Wichtig: Texte immer in Anführungszeichen (") erfassen. Und: Bei Texten wird nicht zwischen Groß- und Kleinschreibung unterschieden.

Beispiel für die Anwendung von ZÄHLENWENN()

Die folgende Abbildung enthält eine Teilnehmerliste für eine Besprechung. In der Spalte B ist ersichtlich, ob die jeweilige Person teilnimmt oder nicht.

Mit einer Formel soll nun in der Zelle D2 die Anzahl der Teilnehmenden berechnet werden. Hierfür müssen lediglich die Zellen in der Spalte B gezählt werden, die den Eintrag „ja“ als Inhalt haben.



	A	B	C	D
1	Name	Teilnahme		Anzahl Teilnehmer
2	Herr Meyer	ja		
3	Frau Müller	nein		
4	Frau Geier	nein		
5	Herr Schmitt	ja		
6	Frau Huber	ja		
7	Herr Becker	ja		
8				
9				

Auswertung von Listen mit der Excel-Funktion ZÄHLENWENN()

Um die Teilnehmenden der Besprechung zu zählen, erfassen Sie einfach die folgende Formel in Zelle D2:

=ZÄHLENWENN(B2:B7; "ja")

Sie bekommen darauf die Anzahl der Teilnehmer in der Zelle D2 dargestellt. Es gibt vier Einträge mit „ja“.

D2					
=ZÄHLENWENN(B2:B7;"ja")					
	A	B	C	D	E
1	Name	Teilnahme		Anzahl Teilnehmer	
2	Herr Meyer	ja		4	
3	Frau Müller	nein			
4	Frau Geier	nein			
5	Herr Schmitt	ja			
6	Frau Huber	ja			
7	Herr Becker	ja			
8					

Ergebnis der Funktion ZÄHLENWENN()

Wollen Sie die Anzahl der Personen zählen, die nicht teilnehmen, dann müssen Sie lediglich das zweite Argument in der Funktion von „ja“ auf „**nein**“ ändern:

=ZÄHLENWENN(B2:B7; "nein")

Excel zählt daraufhin alle Zellen im Bereich B2:B7, die den Text „nein“ als Inhalt haben.

Hinweis: Dabei muss die Schreibweise genau identisch sein. Wird versehentlich **neiin** eingetragen, wird dieser Eintrag nicht mitgezählt. Das ist der Grund, warum es manchmal besser ist, die Einträge in Zellen einzuschränken – zum Beispiel mit einer [Datenüberprüfung und Drop-down-Auswahl](#).

Mit ZÄHLENWENN() nach Zahlen suchen

Um ein Beispiel mit Zahlen als Suchkriterium zu zeigen, wurde das Ausgangsbeispiel in die folgende „Altersliste“ umgewandelt.

	A	B
1	Name	Alter
2	Herr Meyer	50
3	Frau Müller	30
4	Frau Geier	22
5	Herr Schmitt	45
6	Frau Huber	32
7	Herr Becker	55
8		

Liste mit Zahlen für die Auswertung mit ZÄHLENWENN()

In dieser Liste sollen alle Mitarbeitenden gezählt werden, die 50 Jahre alt sind. Hierfür müssen Sie die folgende Formel erfassen:

=ZÄHLENWENN(B2:B7; 50)

Wenn Sie nach einer exakten Zahl suchen, dann erfassen Sie die entsprechende Zahl im Suchkriterium – ohne Anführungszeichen.

D2

▼

:

✕

✓

fx

=ZÄHLENWENN(B2:B7;50)

	A	B	C	D	E
1	Name	Alter		Anzahl	
2	Herr Meyer	50		1	
3	Frau Müller	30			
4	Frau Geier	22			
5	Herr Schmitt	45			
6	Frau Huber	32			
7	Herr Becker	55			
8					

Ergebnis aus dem Zahlenvergleich mit ZÄHLENWENN()

Mit ZÄHLENWENN() Größer- oder Kleiner-Vergleiche durchführen

Nun sollen alle Mitarbeitenden gezählt werden, die jünger als 30 Jahre sind. Sie können die bekannten Vergleichsoperatoren $< = >$ auch für das Suchkriterium in ZÄHLENWENN() einsetzen.

Wichtig ist, dass Sie den Vergleichsoperator und die entsprechende Zahl – analog zu Text – **in Anführungszeichen** erfassen.

Deshalb müssen Sie die folgende Formel erfassen:

=ZÄHLENWENN(B2:B7; "<30")

D2					=ZÄHLENWENN(B2:B7;"<30")
	A	B	C	D	E
1	Name	Alter		Anzahl	
2	Herr Meyer	50		1	
3	Frau Müller	30			
4	Frau Geier	22			
5	Herr Schmitt	45			
6	Frau Huber	32			
7	Herr Becker	55			
8					

ZÄHLENWENN() und Größer- oder Kleiner-Vergleiche

Wenn Sie alle Mitarbeitenden zählen, die 50 oder älter sind, erfassen Sie die folgende Formel:

=ZÄHLENWENN(B2:B7; ">=50")

D2					
=ZÄHLENWENN(B2:B7;">=50")					
	A	B	C	D	E
1	Name	Alter		Anzahl	
2	Herr Meyer	50		2	
3	Frau Müller	30			
4	Frau Geier	22			
5	Herr Schmitt	45			
6	Frau Huber	32			
7	Herr Becker	55			
8					

ZÄHLENWENN() und Größer- oder Kleiner-Gleich-Vergleiche

Hinweis: Bei den Vergleichsoperatoren größer oder gleich (\geq) und kleiner oder gleich (\leq) erfassen Sie das Zeichen = immer an **zweiter Stelle**.

Wenn Sie einen Vergleichsoperator $< = >$ mit einem Verweis auf eine andere Zelle als Suchkriterium in ZÄHLENWENN() einsetzen wollen, dann müssen Sie

- den **Vergleichsoperator** in Anführungszeichen (") erfassen,
- dann ein **&**-Zeichen einfügen und
- dahinter den **Zellverweis** nennen.

Beispiel: In Zelle D2 steht eine Zahl und Sie wollen wissen, wie viele Einträge größer als diese Zahl sind. Die entsprechende Formel lautet dann:

=ZÄHLENWENN(B2:B7; ">"&D2)

Schnittmengen und Vereinigungsmengen mit ZÄHLENWENN() ermitteln

In einem weiteren Fall wollen Sie alle Mitarbeitenden zählen, die älter als 30 und jünger als 50 Jahre alt sind. Mit der Und-Bedingung suchen Sie also die Schnittmenge aus der Altersliste.

Hierfür müssen Sie die folgende Formel erfassen:

=ZÄHLENWENN(B2:B7; ">30") - ZÄHLENWENN(B2:B7; ">=50")

	A	B	C	D	E	F	G
1	Name	Alter		Anzahl			
2	Herr Meyer	50		2			
3	Frau Müller	30					
4	Frau Geier	22					
5	Herr Schmitt	45					
6	Frau Huber	32					
7	Herr Becker	55					

Teilmenge berechnen mit ZÄHLENWENN()

Im Gegensatz zu ZÄHLENWENNS() können Sie derartige Aufgabenstellungen nicht direkt mit ZÄHLENWENN() lösen. Sie müssen hierfür eine Formel mit zwei ZÄHLENWENN()-Funktionen erstellen.

- Mit dem ersten Formelteil **ZÄHLENWENN(B2:B7; ">30")** zählen Sie zunächst alle Mitarbeitenden, die älter als 30 Jahre sind (4 Mitarbeitende).
- Mit dem zweiten Formelteil **ZÄHLENWENN(B2:B7; ">=50")** ermitteln Sie die Anzahl der Mitarbeitenden, die 50 oder älter sind (2 Mitarbeiter).

Dann subtrahieren Sie diese Anzahl vom Ergebnis des ersten Formelteils. Sie schneiden mit der Subtraktion (-) aus der Menge der >30-jährigen die Menge der 50-jährigen und Älteren heraus. Sie erhalten die Anzahl aller Mitarbeitenden, deren Alter zwischen 30 und unter 50 Jahren beträgt.

Entsprechend können Sie mit den Operatoren + und - die Größe von Teilmengen aus Ihrer Liste berechnen. Einfacher geht das aber meist mit der Funktion **ZÄHLENWENNS()**.

Die Excel-Funktion ZÄHLENWENNS()

Wie Sie mit der Funktion ZÄHLENWENNS() umfangreiche Listen nach unterschiedlichen Kriterien durchsuchen und Einträge zählen. Sie bekommen schnell einen Überblick, ob bestimmte Kombinationen vorkommen oder nicht.

Zuletzt geändert am 18.03.2026



Die Funktion **ZÄHLENWENNS()** ist eine Weiterentwicklung der bekannten Funktion **ZÄHLENWENN()** in Excel. Sie zählt zu den statistischen Funktionen, mit denen Sie die Anzahl von Zellen zählen können, die bestimmte Kriterien erfüllen.

Im Gegensatz zu ZÄHLENWENN() können Sie bei ZÄHLENWENNS() **mehrere Kriterien** definieren, die gleichzeitig erfüllt sein müssen, damit die Zellen gezählt werden. Die Kriterien werden also mit **UND** verknüpft.

Folgende Beispiele zeigen, wo und wie die Funktion ZÄHLENWENNS() angewendet wird und was sie leistet.

Aufbau und Funktionsweise von ZÄHLENWENNS()

Die Syntax lautet wie folgt:

=ZÄHLENWENNS(Bereich1; Suchkriterien1; [Bereich2]; [Suchkriterien2] ...)

- **Bereich1**: Wo möchten Sie suchen? Hier erfassen Sie den Bereich, der nach einem bestimmten Kriterium durchsucht werden soll.
- **Suchkriterien1**: Wonach möchten Sie suchen? Erfassen Sie hier das Kriterium, das im zuvor genannten Bereich gesucht und gezählt werden soll.
- **Bereich2** (optional): Sie können weitere Bereiche angeben, die durchsucht werden sollen.
- **Suchkriterien2** (optional): Entsprechend geben Sie ein weiteres Kriterium ein, dessen Vorkommen in Bereich2 gezählt werden soll.


Wichtig: Texte immer in Anführungszeichen ("") erfassen. Bei Texten wird nicht zwischen Groß- und Kleinschreibung unterschieden.

Auf den ersten Blick ist – mit Ausnahme des letzten Buchstabens im Namen der Funktion – kein Unterschied zu ZÄHLENWENN() ersichtlich. Ihre Stärke spielt die Funktion erst dann aus, wenn weitere Kriterien ins Spiel kommen.

Beispiel für die Auswertung einer Liste mit ZÄHLENWENNS()

Die folgende Abbildung enthält eine Liste von eingeladenen Personen für eine Besprechung. In der Spalte B ist ersichtlich, zu welcher Abteilung die jeweilige Person gehört.

Mit einer Formel soll nun in der Zelle E2 die Anzahl der eingeladenen Personen aus der Abteilung K2 berechnet werden. Hierfür müssen lediglich die Zellen in der Spalte B gezählt werden, welche den Eintrag „K2“ als Inhalt haben.



	A	B	C	D	E
1	Name	Abteilung	Teilnahme		Anzahl Teilnehmer
2	Herr Meyer	K1	ja		
3	Frau Müller	K1	nein		
4	Frau Geier	K2	nein		
5	Herr Schmitt	K2	ja		
6	Frau Huber	K3	ja		
7	Herr Becker	K3	ja		

Beispiel für eine Liste zur Auswertung mit der Funktion ZÄHLENWENNS()

Um die Mitarbeitenden aus der Abteilung K2 zu zählen, erfassen Sie einfach die folgende Formel in der Zelle E2:

=ZÄHLENWENNS(B2:B7; "K2")

Sie bekommen darauf die Anzahl der eingeladenen Personen aus der Abteilung K2 in der Zelle E2 dargestellt.

E2					
=ZÄHLENWENNS(B2:B7;"K2")					
	A	B	C	D	E
1	Name	Abteilung	Teilnahme		Anzahl Teilnehmer
2	Herr Meyer	K1	ja		2
3	Frau Müller	K1	nein		
4	Frau Geier	K2	nein		
5	Herr Schmitt	K2	ja		
6	Frau Huber	K3	ja		
7	Herr Becker	K3	ja		
8					

Ergebnis der Funktion ZÄHLENWENNS()

Mehrere Suchkriterien

Der große Vorteil von ZÄHLENWENNS() gegenüber ZÄHLENWENN() ist, dass Sie mit ZÄHLENWENNS() **mehrere Kriterien** definieren können.

Beispiel: Sie wollen nur die Mitarbeitenden aus Abteilung K2 zählen, die ihre Teilnahme an der Besprechung zugesagt haben. Hierfür müssen Sie die folgende Formel erfassen:

=ZÄHLENWENNS(B2:B7; "K2"; C2:C7; "ja")

E2

:

✕

✓

fx

=ZÄHLENWENNS(B2:B7;"K2";C2:C7;"ja")

	A	B	C	D	E	F
1	Name	Abteilung	Teilnahme		Anzahl Teilnehmer	
2	Herr Meyer	K1	ja		1	
3	Frau Müller	K1	nein			
4	Frau Geier	K2	nein			
5	Herr Schmitt	K2	ja			
6	Frau Huber	K3	ja			
7	Herr Becker	K3	ja			
8						

Kombination von Suchbereichen und Suchkriterien mit ZÄHLENWENNS()

Erfassen Sie einfach für die zusätzlichen Kriterien einen weiteren Bereich und entsprechende Suchkriterien, getrennt durch Semikolon.

Wollen Sie noch weitere Kriterien definieren, dann gehen Sie analog vor und erfassen Sie weitere Bereiche und Suchkriterien am Ende der bereits vorhandenen Argumente in der Funktion. Sie können so **bis zu 127 unterschiedliche Bereiche und Suchkriterien** prüfen.

Eingabemöglichkeiten für Suchkriterien

Analog zu ZÄHLENWENN() können Sie auch in ZÄHLENWENNS() **Zahlen** als Suchkriterien verwenden. Erfassen Sie eine **Zahl im Suchkriterium immer ohne Anführungszeichen**.

Wenn Sie hingegen die bekannten Vergleichsoperatoren < = > auch für das Suchkriterium in ZÄHLENWENNS() einsetzen wollen, dann müssen Sie sowohl den Vergleichsoperator als auch die entsprechende Zahl – analog zu Text – in Anführungszeichen erfassen; zum Beispiel: ">=50". Dann werden alle Einträge im Bereich gesucht und gezählt, die größer oder gleich 50 sind.

Sie können bei der **Textsuche mit den Platzhaltern** * und ? arbeiten. Ein Sternchen (*) ersetzt beliebig viele Zeichen; ein Fragezeichen (?) ersetzt genau ein Textzeichen.

Beispiel: Sie suchen in einem Bereich alle Einträge mit „Äpfel“ oder „Apfel“. Entsprechend geben Sie die Formel ein:

=ZÄHLENWENNS(B2:B7; "?pfel")

Sie können als Suchkriterium auch **auf eine andere Zelle verweisen**, in der das Suchkriterium eingetragen ist. So können Sie beispielsweise in Zelle E2 eintragen, nach welcher Abteilung Sie suchen.

In Zelle E3 soll die Zahl der Zusagen aus dieser Abteilung ausgegeben werden. Erfassen Sie dazu in Zelle E3 die Formel:

=ZÄHLENWENNS(B2:B7; E2; C2:C7; "ja")

E3					=ZÄHLENWENNS(B2:B7;E2;C2:C7;"ja")				
	A	B	C	D	E				
1	Name	Abteilung	Teilnahme		Anzahl Teilnehmer				
2	Herr Meyer	K1	ja		K3				
3	Frau Müller	K1	nein		2				
4	Frau Geier	K2	nein						
5	Herr Schmitt	K2	ja						
6	Frau Huber	K3	ja						
7	Herr Becker	K3	ja						
8									
9									

ZÄHLENWENNS() mit Verweis auf Suchkriterium

Wenn Sie einen Vergleichsoperator $< = >$ mit einem Verweis auf eine andere Zelle als Suchkriterium in ZÄHLENWENNS() einsetzen wollen, dann müssen Sie

- den **Vergleichsoperator** in Anführungszeichen (") erfassen,
- dann ein **&**-Zeichen einfügen und
- dahinter den **Zellverweis** nennen.

Beispiel: In Zelle D2 steht eine Zahl und Sie wollen wissen, wie viele Einträge größer als diese Zahl sind. Die entsprechende Formel lautet dann:

=ZÄHLENWENNS(B2:B7; ">"&D2)

Entsprechend können Sie bei ZÄHLENWENNS() mehrere Suchkriterien festlegen und Ihre Daten auswerten.

Zellen zählen in Excel – mit ODER-Bedingung

Wie viele Zellen in einem Datensatz enthalten den Begriff „Apfel“ oder „Orange“? Eine solche kombinierte Suche mit ODER-Bedingung lässt sich mit dieser Excel-Formel schnell erledigen.

Zuletzt geändert am 18.03.2026



Bei der Datenanalyse und in großen Datenmengen, etwa in Kundenlisten, Produktdatenbanken oder Inventarübersichten, müssen oft bestimmte Werte oder Kombinationen schnell identifiziert werden.

Eine Aufgabe kann sein, dass Sie **Zellen zählen** wollen, die **mehr als eine Bedingung erfüllen** – zum Beispiel Zellen, die entweder den einen oder einen anderen bestimmten Inhalt haben.

Mithilfe von Excel-Formeln können Sie flexibel und einfach Zellen zählen, die mehrere Kriterien gleichzeitig erfüllen.

Daten auswerten und Zellen zählen mit ODER-Bedingung

In der folgenden Abbildung sollen alle Zellen gezählt werden, die den Text „Apfel“ **oder** „Orange“ beinhalten. Die Groß- und Kleinschreibung soll hierbei nicht beachtet werden.

Es handelt sich somit um eine klassische **ODER-Bedingung**, für die Sie folgende Formel anwenden:

```
{=SUMME(WENN(ISTZAHL(SUCHEN("Apfel"; A1:A10)); 1;  
WENN(ISTZAHL(SUCHEN("Orange"; A1:A10)); 1; 0)))}
```

Die Formel geht hierbei von den folgenden Voraussetzungen aus:

- **Apfel:** erster Suchbegriff
- **Orange:** zweiter Suchbegriff
- **A1:A10:** Bereich, in dem die beiden Suchbegriffe gesucht werden

Hinweis: Erfassen Sie die Formel ohne die geschweiften Klammern am Anfang und am Ende. Die Formel bestätigen Sie nicht mit Enter, sondern mit der Tastenkombination **Strg + Umschalt + Enter**, da es sich um eine **Matrixformel** handelt. Die geschweiften Klammern werden durch die Tastenkombination automatisch eingefügt.

The screenshot shows an Excel spreadsheet with the following data:

	A	B	C	D	E	F	G	H	I
1	orange		5						
2	Zitrone								
3	Zitrone								
4	Apfel								
5	Apfel								
6	Orange								
7	Traube								
8	apfel								
9	Zitrone								
10	Traube								

The formula bar shows the following formula: `=SUMME(WENN(ISTZAHL(SUCHEN("Apfel";A1:A10));1;WENN(ISTZAHL(SUCHEN("Orange";A1:A10));1;0)))`

Formel für das Zählen von Dateneinträgen mit ODER-Bedingung

Mit der Funktion **SUCHEN("Apfel"; A1:A10)** wird zunächst der Bereich A1:A10 nach dem Begriff „Apfel“ durchsucht. Die Groß- und Kleinschreibung wird hierbei nicht beachtet.

- Kann der Begriff „Apfel“ gefunden werden, dann gibt die Funktion SUCHEN() die Position als Ergebnis zurück.
- Kann der Begriff nicht gefunden werden, dann wird der Fehlerwert #WERT! als Ergebnis ausgegeben.

Im nächsten Schritt wird das Ergebnis an die Funktion ISTZAHL() übergeben:

ISTZAHL(SUCHEN("Apfel"; A1:A10)). Hier wird überprüft, ob das Ergebnis eine Zahl ist (Suchbegriff wurde gefunden) oder nicht (Suchbegriff wurde nicht gefunden).

Die Funktion ISTZAHL() ist in eine WENN-Funktion verschachtelt. Sollte eine Zahl als Ergebnis vorliegen, dann wird für die entsprechende Zelle der Wert 1 über die WENN-Funktion zurückgemeldet: **WENN(ISTZAHL(SUCHEN("Apfel"; A1:A10)); 1**.

Sollte keine Zahl vorliegen, dann greift das Sonst-Argument in der WENN-Funktion. Hier wird nun mit dem analogen Formelteil, nach dem Suchbegriff „Orange“ gesucht:

```
{=SUMME(WENN(ISTZAHL(SUCHEN("Apfel"; A1:A10));  
1; WENN(ISTZAHL(SUCHEN("Orange"; A1:A10)); 1; 0)))}
```

Sollte der Begriff „Orange“ gefunden werden, dann wird der Wert 1 sonst der Wert 0 über die WENN-Funktion als Ergebnis zurückgegeben.

Die Ergebnisse (0 oder 1) der beiden WENN-Funktionen für jede Zelle im Bereich A1:A10 werden anschließend an die Funktion SUMME() übergeben, die dann die einzelnen Werte summiert.

```
{=SUMME(WENN(ISTZAHL(SUCHEN("Apfel"; A1:A10)); 1;  
WENN(ISTZAHL(SUCHEN("Orange"; A1:A10)); 1; 0)))}
```

Sie erhalten somit die Anzahl der Zellen im Bereich A1:A10, die den Begriff „Apfel“ oder „Orange“ beinhalten. Sollten die Begriffe „Apfel“ und „Orange“ beide in einer einzelnen Zelle vorkommen, dann wird diese Zelle nur einmal gezählt.

Schreibweise bei der Suche berücksichtigen mit FINDEN()

Soll die Groß- und Kleinschreibung berücksichtigt werden, dann müssen Sie lediglich die SUCHEN()-Funktion in der Formel durch die **FINDEN()-Funktion** ersetzen:

```
{=SUMME(WENN(ISTZAHL(FINDEN("Apfel"; A1:A10)); 1;  
WENN(ISTZAHL(FINDEN("Orange"; A1:A10)); 1; 0)))}
```

	A	B	C	D	E	F	G	H	I
1	orange		3						
2	Zitrone								
3	Zitrone								
4	Apfel								
5	Apfel								
6	Orange								
7	Traube								
8	apfel								
9	Zitrone								
10	Traube								

Formel für das Zählen von Dateneinträgen mit ODER-Bedingung und Beachtung der Schreibweise

Ungerade Werte in einer Excel-Liste zählen

Erfahren Sie, wie Sie die ungeraden Werte in einer Zahlenliste mit einer einfachen Formel zählen. Und mit einer einfachen Anpassung der Formel zählen Sie auf die gleiche Weise die geraden Werte in Ihrer Liste.

Zuletzt geändert am 18.03.2026



Die **Auswahl bestimmter Werte** aus einer Liste ist eine immer wiederkehrende Aufgabe für Excel. Vieles gelingt nur durch geschickte Abfragen oder eine Filterfunktion.

Eine Aufgabe kann beispielsweise darin bestehen, gerade oder ungerade Zahlen in einer Liste zu identifizieren und zu zählen.

Das Zählen von geraden oder ungeraden Zahlen ist in der Praxis weitverbreitet und zählt zum Beispiel zu den Standardaufgaben in der Qualitätskontrolle oder der Finanzanalyse. Leider gibt es hierfür keine Standardfunktion in Excel.

Formel für das Zählen der ungeraden Zahlen in einer Liste

Mit der folgenden Formel können Sie alle ungeraden Zahlen aus einem Bereich zählen:

`{=SUMME(WENN(REST(A2:A11; 2)<>0; 1; 0))}`

Die Formel geht dabei davon aus, dass die Zahlenliste im Bereich A2:A11 ausgewertet werden soll.

Erfassen Sie die Formel ohne die geschweiften Klammern am Anfang und am Ende. Bestätigen Sie die Formel nicht mit Enter, sondern mit der Tastenkombination **Strg + Umschalt + Enter**, da es sich um eine **Matrixformel** handelt. Die geschweiften Klammern werden durch die Tastenkombination automatisch eingefügt.

Hinweis: Bei neueren Excelversionen können Sie die Formel auch mit Enter bestätigen.

C2			{=SUMME(WENN(REST(A2:A11;2)<>0;1;0))}		
	A	B	C	D	E
1	Werte		Ungerade Zahlen		
2	2		3		
3	2				
4	5				
5	5				
6	2				
7	2				
8	1				
9	2				
10	2				
11	2				

Formel für das Zählen der ungeraden Werte

Welche Funktionen stecken in der Formel?

Zunächst wird mit der Funktion **REST()** jeder Wert im Bereich **A2:A11** durch **2** dividiert.

```
{=SUMME(WENN(REST(A2:A11; 2)<>0; 1; 0))}
```

Durch die Funktion **REST()** wird ermittelt, ob beim Division-Vorgang ein Rest übrigbleibt oder nicht.

Beispiel: Bei der Division von 2 durch 2 bleibt der Rest 0 übrig. Es handelt sich somit um eine **gerade Zahl**. Bei der Division von 5 durch 2 bleibt ein Rest von 1 übrig ($5 \div 2 = 2$ Rest 1). Es handelt sich somit um eine **ungerade Zahl**.

Dann wird mit der **WENN**-Funktion verglichen, ob der Rest ungleich dem Wert 0 ist.

```
{=SUMME(WENN(REST(A2:A11; 2)<>0; 1; 0))}
```

Ist der Wert **ungleich 0**, dann handelt es sich um eine **ungerade Zahl** und es wird über die WENN-Funktion der Wert 1, sonst 0 (für gerade Zahl) ausgegeben.

```
{=SUMME(WENN(REST(A2:A11; 2)<>0; 1; 0))}
```

Anschließend werden alle Werte der WENN-Funktion an die Funktion SUMME() übergeben, welche die Werte summiert und somit alle ungeraden Zahlen in dem entsprechenden Bereich zählt.

```
{=SUMME(WENN(REST(A2:A11; 2)<>0; 1; 0))}
```

Gerade Zahlen in der Liste zählen

Wollen Sie alle geraden Zahlen in einer Liste zählen, dann müssen Sie die Formel nur wie folgt anpassen:

```
{=SUMME(WENN(REST(A2:A11; 2)=0; 1; 0))}
```

Im Gegensatz zu den ungeraden Zahlen werden jetzt die Zahlen gesucht, bei denen kein Rest bei der Division durch 2 übrigbleibt. Sie müssen also nur den Vergleichsoperator in der WENN-Funktion entsprechend anpassen.

C2			$\{=SUMME(WENN(REST(A2:A11;2)=0;1;0))\}$		
	A	B	C	D	E
1	Werte		Gerade Zahlen		
2	2		7		
3	2				
4	5				
5	5				
6	2				
7	2				
8	1				
9	2				
10	2				
11	2				

Formel für das Zählen der geraden Werte

Unterschiedliche Werte in einer Excel-Liste zählen

Wie Sie umfangreiche Listen auswerten und die Anzahl der unterschiedlichen Einträge zählen. Doppelte und Mehrfachnennungen werden nur einmal mitgezählt. So gehen Sie vor.

Zuletzt geändert am 18.03.2026



Wenn Sie mit großen Datenmengen in Excel arbeiten, stehen Sie möglicherweise vor der Herausforderung, die **Anzahl der unterschiedlichen Werte in einer Liste** zu ermitteln.

Ganz gleich, ob Sie doppelte Einträge in einer Kundenliste vermeiden oder die Vielfalt der Produkte in einem Bestandsbericht überprüfen möchten – die Möglichkeit, unterschiedliche Werte zu zählen, ist ein unverzichtbares Werkzeug für jeden Excel-Anwender.

In diesem Excel-Tipp erfahren Sie, wie Sie mit einfachen und fortgeschrittenen Excel-Funktionen die Anzahl der eindeutigen oder unterschiedlichen Elemente in einer Liste bestimmen können.

Beispiel: Liste mit mehrfach auftretenden Einträgen

In der folgenden Tabelle sehen Sie eine Liste von Namen. Manche Namen kommen mehrfach vor, während einige Namen nur einmal aufgelistet sind.

Aufgabe ist es, die Anzahl der eindeutigen Werte zu berechnen. Jeder Wert in der Liste soll nur einmal gezählt werden. Kommt ein Wert zweimal oder öfter vor, soll er nur einmal gezählt werden.

	A	
1	Name	
2	Müller	
3	Meier	
4	Müller	
5	Schmitt	
6	Schmidt	
7	Maier	
8	Müller	
9	Müller	
10	Meier	

Liste mit mehrfachen Einträgen

Die Anzahl der unterschiedlichen Werte in einer Liste können Sie mit der folgenden Formel berechnen:

=ANZAHL2(EINDEUTIG(A2:A10))

Der Bereich A2:A10 repräsentiert den Bereich, in dem die Daten stehen.

D2				
=ANZAHL2(EINDEUTIG(A2:A10))				
	A	B	C	D
1	Name			
2	Müller		Anzahl eindeutiger Werte	5
3	Meier			
4	Müller			
5	Schmitt			
6	Schmidt			
7	Maier			
8	Müller			
9	Müller			
10	Meier			

So zählen Sie eindeutige Werte in einer Liste

Funktion EINDEUTIG()

Mit der Funktion **EINDEUTIG(A2:A10)** werden zunächst die mindestens einmal vorkommenden Werte in der Liste ermittelt. Mehrfachnennungen werden hierbei nicht beachtet. Die Formel liefert für das Beispiel die folgende Auflistung:

- Müller
- Meier
- Schmitt
- Schmidt
- Maier

Hinweis: Kommen in der Liste leere Zellen vor, dann liefert EINDEUTIG() für diese Zellen den Wert 0.

Funktion ANZAHL2()

Die Auflistung wird anschließend an die Funktion **ANZAHL2()** übergeben, die dann die Anzahl der Einträge zählt.

Mit der Funktion **ANZAHL2()** werden Zellen ermittelt, die beliebige Arten von Informationen enthalten. Das können auch Fehlerwerte oder leerer Text sein, wenn die Zelle beispielsweise eine Formel enthält, die eine leere Zeichenfolge ("") zurückgibt.

In der Kombination mit EINDEUTIG() wertet ANZAHL2() die Liste mit den genau einmal auftretenden Einträgen aus und zählt, wie viele dies sind. Da EINDEUTIG() für leere Zellen eine 0 ausgibt, zählt ANZAHL2() diese mit.

Hinweis: Wenn Sie nur Zellen berücksichtigen möchten, die Zahlen enthalten, verwenden Sie die Funktion **ANZAHL()**.

Alternative Formel für die Funktion EINDEUTIG()

Setzen Sie eine ältere Version von Excel ein, dann kann es sein, dass die Funktion EINDEUTIG() nicht funktioniert.

In diesem Fall können Sie die folgende Formel einsetzen:

{=SUMME(1/ZÄHLENWENN(A2:A10; A2:A10))}

Der Bereich A2:A10 repräsentiert den Bereich, in dem die Daten stehen.

Hinweis: Erfassen Sie die Formel **ohne die geschweiften Klammern** am Anfang und am Ende. Bestätigen Sie die Formel nicht mit Enter, sondern mit **Strg + Umschalt + Enter**, da es sich um eine **Matrixformel** handelt. Die geschweiften Klammern werden daraufhin automatisch am Anfang und am Ende eingefügt.

D2 ✕ ✓ <i>fx</i> {=SUMME(1/ZÄHLENWENN(A2:A10;A2:A10))}				
	A	B	C	D
1	Name			
2	Müller		Anzahl eindeutiger Werte	5
3	Meier			
4	Müller			
5	Schmitt			
6	Schmidt			
7	Maier			
8	Müller			
9	Müller			
10	Meier			

Zählen eindeutiger Werte mit Matrixformel

Die Matrixformel funktioniert wie folgt: Zuerst wird mit der Formel **ZÄHLENWENN(A2:A10; A2:A10)** überprüft, wie oft jeder einzelne Wert in der Liste vorkommt.

{=SUMME(1/ZÄHLENWENN(A2:A10; A2:A10))}

Anschließend wird der Wert 1 durch die festgestellte Anzahl an Vorkommen in der Liste dividiert. Kommt ein Wert nur einmal in der Liste vor, dann wird 1 durch 1 dividiert, was zu dem Ergebnis 1 führt.

{=SUMME(1/ZÄHLENWENN(A2:A10; A2:A10))}

Der Name Müller kommt in der Beispielliste viermal vor. Das bedeutet, dass der Wert 1 durch 4 dividiert wird. Das Ergebnis beträgt 0,25. Da der Wert Müller viermal in der Liste vorkommt, wird viermal der Wert 0,25 durch die Division ermittelt. In der Summe kommen Sie somit bei dem Namen Müller auf den Wert 1 ($4 \times 0,25 = 1$).

Alle durch die Division ermittelten Werte werden anschließend an die Funktion SUMME() übergeben, welche die einzelnen Werte summiert. Sie erhalten als Ergebnis die Anzahl der unterschiedlichen Werte in der Liste.

{=SUMME(1/ZÄHLENWENN(A2:A10; A2:A10))}

Werte mit Excel außerhalb der Toleranz zählen (Toleranzformel)

Mit einer einfachen Formel können Sie in Excel Messwerte auswerten und ermitteln, wie viele Werte außerhalb eines vorgegebenen Toleranzbereichs liegen. Das ist hilfreich bei der Qualitätskontrolle.

Zuletzt geändert am 18.03.2026



Beispiel: Liste mit Messwerten, Zielwert und Toleranz

Um Werte zu zählen, die in einer Liste außerhalb einer bestimmten Toleranz liegen, können Sie eine **Matrixformel** einsetzen.

Sie sehen in der folgenden Abbildung eine Liste von Messwerten. In Zelle B2 ist der Zielwert (1,00) eingetragen. Erlaubt ist eine Toleranzabweichung von 0,05 nach oben und unten, was in Zelle C2 eingetragen ist.

	A	B	C	
1	Wert	Zielwert	Toleranz	
2	0,91	1,00	0,05	
3	0,95			
4	0,92			
5	1,01			
6	0,90			
7	1,00			
8	1,03			
9	0,95			
10	0,93			
11	1,07			
12	1,01			
13	1,09			
14	1,01			
15	0,90			
16	1,01			
17	1,06			
18	1,02			
19	0,93			
20	1,04			
21	1,05			

Liste mit Messwerten und Vorgabe von Zielwert und Toleranzbereich

Formel für das Zählen von Werten außerhalb eines Toleranzbereichs

Um alle Werte zu zählen, die sich außerhalb der Toleranz befinden, können Sie die folgende Formel einsetzen:

{=SUMME(WENN(ABS(B2-A2:A21)>C2; 1; 0))}

Hinweis: Erfassen Sie die Formel ohne die geschweiften Klammern am Anfang und am Ende. Bestätigen Sie die Formel dann aber nicht mit Enter, sondern mit der

Tastenkombination **Strg + Umschalttaste + Enter**, da es sich um eine Matrixformel handelt. Durch die Tastenkombination werden dann automatisch die geschweiften Klammern am Anfang und Ende eingefügt. Besitzen Sie ein **Microsoft-365-Abo**, dann können Sie die Formel auch normal mit **Enter** erfassen.

E2	{=SUMME(WENN(ABS(B2-A2:A21)>C2;1;0))}				
	A	B	C	D	E
1	Wert	Zielwert	Toleranz		außerhalb der Toleranz
2	0,91	1,00	0,05		9
3	0,95				
4	0,92				
5	1,01				
6	0,90				
7	1,00				
8	1,03				
9	0,95				
10	0,93				
11	1,07				
12	1,01				
13	1,09				
14	1,01				
15	0,90				
16	1,01				
17	1,06				
18	1,02				
19	0,93				
20	1,04				
21	1,05				

Anzahl der Werte außerhalb des Toleranzbereichs

Die Werte, die außerhalb des Toleranzbereichs liegen, sind in der Abbildung mit einer bedingten Formatierung hervorgehoben. Geprüft wird: Welcher **Zelleintrag** ist **größer als \$B\$2+\$C\$2** oder **kleiner als \$B\$2-\$C\$2**?

Was berechnet die Formel?

Die Matrixformel ermittelt zuerst für jede Zelle im Bereich A2:A21 die Abweichung zum Zielwert in Zelle B2 (1,00): =SUMME(WENN(ABS(B2-A2:A21)>C2; 1; 0))

Anschließend wird die Abweichung pro Zelle an die Funktion **ABS()** weitergeleitet, die aus dem Wert den **absoluten Wert** berechnet.

Der absolute Wert einer Zahl ist der Betrag der Zahl, unabhängig von ihrem Vorzeichen. Er gibt an, wie weit die Zahl von Null entfernt ist. Der absolute Wert einer Zahl macht somit eine negative Zahl positiv.

Nach dieser Berechnung gibt es somit nur noch positive Werte (Abweichungen), weil es für die Aufgabenstellung unerheblich ist, ob die Abweichung nach oben oder nach unten erfolgt.

```
{=SUMME(WENN(ABS(B2-A2:A21)>C2; 1; 0))}
```

Die (positiven) Abweichungen werden nun an die WENN()-Funktion weitergeleitet. Ist die Abweichung größer als die Toleranz in Zelle C2 (0,05), dann gibt die WENN()-Funktion den Wert 1, sonst 0 zurück.

```
{=SUMME(WENN(ABS(B2-A2:A21)>C2; 1; 0))}
```

Die Werte der WENN()-Funktion – also alle 0er und 1er – werden anschließend an die Funktion SUMME() weitergeleitet, welche die Werte addiert.

```
{=SUMME(WENN(ABS(B2-A2:A21)>C2; 1; 0))}
```

Auf diese Weise können Sie alle Messungen in einer Liste zählen, deren Wert außerhalb der Zieltoleranz ist.

Werte innerhalb des Toleranzbereichs zählen

Wollen Sie hingegen die Werte zählen, die sich innerhalb der Toleranzabweichung befinden, dann müssen Sie die Formel nur geringfügig anpassen. Tauschen Sie einfach in der WENN-Funktion den Vergleichsoperator > durch <= aus:

```
{=SUMME(WENN(ABS(B2-A2:A21)<=C2; 1; 0))}
```

E2 {=SUMME(WENN(ABS(B2-A2:A21)<=C2;1;0))}					
	A	B	C	D	E
1	Wert	Zielwert	Toleranz		innerhalb der Toleranz
2	0,91	1,00	0,05		11
3	0,95				
4	0,92				
5	1,01				
6	0,90				
7	1,00				
8	1,03				
9	0,95				
10	0,93				
11	1,07				
12	1,01				
13	1,09				
14	1,01				
15	0,90				
16	1,01				
17	1,06				
18	1,02				
19	0,93				
20	1,04				
21	1,05				

Anzahl der Messwerte innerhalb des Toleranzbereichs

Hinweis: In diesem Fall sind die Werte mit einer **Bedingten Formatierung** hervorgehoben, die innerhalb des Toleranzbereichs liegen.



Eine entsprechende Auswertung von Messwerten ist die Grundlage der Qualitätskontrolle. Weitere Informationen und Excel-Tools finden Sie im **Beitrag zur Qualitätsregelkarte**.

Werte mit Excel außerhalb eines Intervalls zählen

So zählen Sie die Anzahl der Zellen, deren Werte nicht innerhalb eines bestimmten Intervalls liegen. Mit den Funktionen ZÄHLENWENN() und SUMMENPRODUKT() prüfen Sie die Spezifikation und zählen dann die Anzahl der Werte, die sie erfüllen.

Zuletzt geändert am 18.03.2026



In vielen Datenanalysen – etwa bei der Auswertung von Verkaufszahlen, Temperaturen oder Qualitätskontrollen – stehen Sie möglicherweise vor der Aufgabe, **Zellen** zu zählen, deren Werte **außerhalb eines bestimmten Intervalls** liegen.

In diesem Beitrag erfahren Sie, wie Sie diese Anforderung mit einer einfachen und dynamischen Formel in Excel lösen können.

Beispiel: Qualitätskontrolle in der Produktion

Angenommen, Sie arbeiten in der Produktionskontrolle und haben die Messdaten zur Größe von Bauteilen in Millimetern in einer Excel-Tabelle gespeichert. Die Bauteile sollten idealerweise zwischen 20 mm und 80 mm groß sein.

Sie möchten schnell erkennen, wie viele Bauteile dieser Spezifikation nicht entsprechen, um entsprechende Maßnahmen einzuleiten.

Die folgende Tabelle beinhaltet die Messwerte der Bauteile. Es soll die Anzahl der Bauteile gezählt werden, deren Messwerte außerhalb des Intervalls von 20 mm bis 80 mm liegen. Die entsprechenden Werte sind in der Tabelle bereits farblich hervorgehoben.

	A	B	C	D	E	F
1	10	9	93	95	19	18
2	70	37	3	87	45	40
3	70	45	13	27	65	18
4	41	39	71	20	32	52
5	5	95	73	22	52	50
6	32	54	62	66	84	71
7	1	30	45	79	15	28
8	26	10	88	93	37	3
9	30	74	24	68	62	4
10	19	100	19	19	53	73
11	45	65	50	69	36	19
12	28	90	37	27	22	88
13	78	14	47	24	80	61
14	8	35	85	35	50	84
15	83	57	14	27	53	58

Messwerte zur Auswertung von Spezifikationen

Die Lösung: ZÄHLENWENN-Funktion verwenden

Mit der ZÄHLENWENN-Funktion können Sie ganz einfach Bedingungen definieren, um Zellen zu zählen, die nicht innerhalb eines Intervalls liegen.

Wenn die Grenzen des Intervalls 20 und 80 betragen, verwenden Sie die folgende Formel:

=ZÄHLENWENN(A1:F15; "<20") + ZÄHLENWENN(A1:F15; ">80")

A1:A15 ist der Bereich, in dem sich die Werte befinden.

G1 : \times \checkmark f_x =ZÄHLENWENN(A1:F15;"<20")+ZÄHLENWENN(A1:F15;">80")							
	A	B	C	D	E	F	G
1	10	9	93	95	19	18	33
2	70	37	3	87	45	40	
3	70	45	13	27	65	18	
4	41	39	71	20	32	52	
5	5	95	73	22	52	50	
6	32	54	62	66	84	71	
7	1	30	45	79	15	28	
8	26	10	88	93	37	3	
9	30	74	24	68	62	4	
10	19	100	19	19	53	73	
11	45	65	50	69	36	19	
12	28	90	37	27	22	88	
13	78	14	47	24	80	61	
14	8	35	85	35	50	84	
15	83	57	14	27	53	58	

Anzahl der Werte außerhalb eines Intervalls berechnen mit ZÄHLENWENN()

Wie funktioniert die Formel?

ZÄHLENWENN(A1:F15; "<20") zählt die Zellen, deren Werte kleiner als 20 sind.

ZÄHLENWENN(A1:F15 ;">80") zählt die Zellen, deren Werte größer als 80 sind.

Anschließend werden beide Ergebnisse addiert, sodass Sie die Gesamtanzahl der Zellen außerhalb des Intervalls erhalten.

Alternative mit SUMMENPRODUKT()

Eine Alternative zu ZÄHLENWENN() ist die Funktion **SUMMENPRODUKT()**. Mit dieser Funktion können Sie die Werte, die außerhalb eines bestimmten Intervalls liegen, auch einfach zählen:

=SUMMENPRODUKT((A1:F15<20) + (A1:F15>80))

G1 : \times \checkmark f_x =SUMMENPRODUKT((A1:F15<20)+(A1:F15>80))							
	A	B	C	D	E	F	G
1	10	9	93	95	19	18	33
2	70	37	3	87	45	40	
3	70	45	13	27	65	18	
4	41	39	71	20	32	52	
5	5	95	73	22	52	50	
6	32	54	62	66	84	71	
7	1	30	45	79	15	28	
8	26	10	88	93	37	3	
9	30	74	24	68	62	4	
10	19	100	19	19	53	73	
11	45	65	50	69	36	19	
12	28	90	37	27	22	88	
13	78	14	47	24	80	61	
14	8	35	85	35	50	84	
15	83	57	14	27	53	58	

Anzahl der Werte außerhalb eines Intervalls berechnen mit SUMMENPRODUKT()

Auch mit dieser Formel werden zunächst alle Zellen gezählt, deren Wert kleiner als 20 ist: SUMMENPRODUKT((**A1:F15<20**) + (A1:F15>80)).

Anschließend werden die Zellen gezählt, deren Wert größer als 80 ist:
=SUMMENPRODUKT((A1:F15<20) + (**A1:F15>80**)).

Danach werden die beiden Werte addiert, sodass Sie die Anzahl der Zellen erhalten, deren Werte außerhalb des Intervalls liegen.

Zellen in Excel in Abhängigkeit der Hintergrundfarbe zählen oder summieren

Mit der Funktion `ZELLE.ZUORDNEN()` aus dem Excel-Makro-Katalog können Sie Zellen abhängig von ihrer Hintergrundfarbe in Formeln nutzen. So zählen oder addieren Sie beispielsweise Werte, die in gelben Zellen stehen.

Zuletzt geändert am 18.03.2026



Oft werden wichtige Zellen, Spalten oder Zeilen in Excel mit Farben markiert, um die Übersichtlichkeit zu verbessern. Können diese Farben auch in Formeln und Funktionen genutzt werden?

Standardmäßig können Sie die Hintergrundfarben in Excel beim **Filtern oder Sortieren** von Daten verwenden. Wollen Sie die Zellen aufgrund ihrer Hintergrundfarbe in Formel oder Funktionen verarbeiten, dann ist dies im Excel-Standard nicht möglich.

Mit einer **alten Excel4-Makrofunktion** können Sie Hintergrundfarben in Formeln und Funktionen einsetzen. Der Name dieser Excel4-Makrofunktion lautet **ZELLE.ZUORDNEN()**.

Was sind Excel4-Makrofunktionen?

Excel4-Makrofunktionen, auch als **XLM-Makrofunktionen** bekannt, stammen aus der Zeit vor der Einführung von VBA (Visual Basic for Applications). Diese Funktionen wurden in Excel 4.0 eingeführt und ermöglichen die Automatisierung von Aufgaben und die Erstellung komplexer Makros in Excel.

Obwohl Excel-4-Makros veraltet sind, unterstützt Microsoft Excel sie weiterhin für die **Rückwärtskompatibilität**. Benutzer, die alte Arbeitsmappen mit XLM-Makros öffnen, können diese weiterhin ausführen, obwohl neue Entwicklungen und Support hauptsächlich auf VBA fokussiert sind.

Sicherheitseinstellungen zu Excel-4-Makrofunktionen


Aufgrund ihrer Fähigkeit, tief in das System einzugreifen und Änderungen vorzunehmen, werden Excel-4-Makros oft als Sicherheitsrisiko betrachtet. Böartige Akteure können diese Makros ausnutzen, um Schadcode auszuführen.

Daher sind viele Sicherheitseinstellungen und Schutzmaßnahmen in modernen Excel-Versionen implementiert, um die Ausführung von XLM-Makros zu überwachen und zu kontrollieren.

Damit Sie die Excel-4-Makrofunktionen nutzen können, müssen Sie zunächst die Excel-Sicherheitseinstellungen überprüfen und gegebenenfalls anpassen.

Öffnen Sie die **Excel-Optionen**, indem Sie im Menüband die Befehlsfolge Registerkarte **Datei** > Befehl **Optionen** ausführen. Daraufhin öffnet sich das Dialogfeld **Excel-Optionen**.

Wählen Sie hier am linken Rand das Register **Trust Center** und anschließend **Einstellungen für das Trust Center** aus.




Sicherheitseinstellungen anpassen in Excel

Es öffnet sich das Dialogfenster **Trust Center**. Wählen Sie hier am linken Rand das Register **Makroeinstellungen** aus.

Überprüfen Sie hier, ob das Kontrollkästchen **Excel 4,0-Makros aktivieren, wenn VBA-Makros aktiviert sind** aktiviert ist.

Sollte dies nicht der Fall sein, dann aktivieren Sie das Kontrollkästchen. Bestätigen Sie diese Einstellung, indem Sie alle offenen Dialogfelder durch einen Klick auf **OK** schließen.




Excel4-Makros zulassen

Namen definieren und verwenden

Sie können die Excel-4-Makrofunktion **ZELLE.ZUORDNEN()** nicht direkt in einem Tabellenblatt einsetzen.

Der erste Schritt, um **Farbeinstellungen von Zellen in Formeln zu nutzen**, besteht darin, einen **Namen zu definieren**, in welchem die Excel-4-Makrofunktion ZELLE.ZUORDNEN() hinterlegt wird.

Aktivieren Sie hierzu im Menüband die Befehlsfolge Registerkarte **Formeln** > Befehlsgruppe **Definierte Namen** > Befehl **Namen definieren**.



Benutzerdefinierte Namen in Excel eingeben


Es öffnet sich das Dialogfeld **Neuer Name**. Vergeben Sie hier bei **Name:** einen aussagekräftigen Namen. Im Beispiel ist dies der Name **Hintergrundfarbe**.

Bei **Bereich** lassen Sie die Auswahl **Arbeitsmappe** stehen. Damit stellen Sie sicher, dass der definierte Name in der kompletten Arbeitsmappe und nicht nur in einem bestimmten Tabellenblatt verwendet werden kann.

Unter **Bezieht sich auf:** erfassen Sie die folgende Formel:

=ZELLE.ZUORDNEN(63,INDIREKT("ZS(-1)");))

Bestätigen Sie Ihre Einstellung, indem Sie das Dialogfeld durch einen Klick auf **OK** schließen.



Funktion ZELLE.ZUORDNEN() als Name hinterlegen

Mit der Excel-4-Makrofunktion `ZELLE.ZUORDNEN()` kann man Eigenschaften einer Zelle abfragen. Das erste Argument **63** besagt, dass Sie die **Hintergrundfarbe** (Farbcode) einer Zelle auslesen wollen.

Mit dem zweiten Argument **INDIREKT("ZS(-1)");** geben Sie die Position der Zelle an, aus welcher Sie die Hintergrundfarbe auslesen wollen.

In der `INDIREKT`-Funktion geben Sie die Zelladresse in der **Z1S1-Schreibweise** an. In dieser Schreibweise werden nicht Koordinaten verwendet, sondern Zeilen- und Spaltennummern.

So wird aus der Zelle **A1** eben **Z1S1** und aus der Zelle **B5** wird **Z5S2**. Die oben verwendete Angabe **ZS(-1)** verweist damit auf die Zelle links neben sich.

Verwendung der definierten Namen in Excel-Formeln und Berechnungen

Nach dem Definieren des Namens können Sie diesen in einer Zelle verwenden, um die entsprechende Farbe zu extrahieren.

In der folgenden Abbildung sehen Sie eine Tabelle, in der Umsätze mit einer Hintergrundfarbe hervorgehoben sind. Sie wollen diese Umsätze anhand der Hintergrundfarben nun summieren oder zählen.

	A	B	
1	Produkt 1	1.466	
2	Produkt 2	1.652	
3	Produkt 3	1.376	
4	Produkt 4	1.944	
5	Produkt 5	1.169	
6	Produkt 6	1.329	
7	Produkt 7	1.067	
8	Produkt 8	1.189	
9	Produkt 9	1.397	
10	Produkt 10	1.128	
11			

Liste mit Werten und Hintergrundfarbe

Schreiben Sie in eine Zelle rechts neben der ersten Umsatzzahl die Formel:

=Hintergrundfarbe

Excel zeigt dann die entsprechende Zahl (Farbcode) für die Hintergrundfarbe an, die in der Zelle links daneben hinterlegt ist.

	A	B	C	D
1	Produkt 1	1.466	6	
2	Produkt 2	1.652		
3	Produkt 3	1.376		
4	Produkt 4	1.944		
5	Produkt 5	1.169		
6	Produkt 6	1.329		
7	Produkt 7	1.067		
8	Produkt 8	1.189		
9	Produkt 9	1.397		
10	Produkt 10	1.128		
11				

Wert für die Hintergrundfarbe ermitteln

Ziehen oder kopieren Sie die Formel nach unten, um die Farben (Farbcodes) der anderen Zellen auszulesen.

	A	B	C	D
1	Produkt 1	1.466	6	
2	Produkt 2	1.652	6	
3	Produkt 3	1.376	44	
4	Produkt 4	1.944	0	
5	Produkt 5	1.169	33	
6	Produkt 6	1.329	3	
7	Produkt 7	1.067	6	
8	Produkt 8	1.189	6	
9	Produkt 9	1.397	6	
10	Produkt 10	1.128	6	
11				
12				

Liste mit den Werten aller Hintergrundfarben

Mit diesen Zahlen (Farbcodes) in der Hilfsspalte können Sie nun weitere Berechnungen durchführen, zum Beispiel die Anzahl der Zellen mit einer bestimmten Farbe zählen.

Beispiel: Zählen aller gelben Zellen

Um die Anzahl der Zellen mit einer bestimmten Hintergrundfarbe zu zählen, verwenden Sie die **ZÄHLENWENN-Funktion**. Zählen Sie einfach, wie oft der Farbcode 6 (gelber Hintergrund) in der Hilfsspalte C vorkommt.

C12 ✕ ✓ fx =ZÄHLENWENN(C1:C10;6)					
	A	B	C	D	
1	Produkt 1	1.466	6		
2	Produkt 2	1.652	6		
3	Produkt 3	1.376	44		
4	Produkt 4	1.944	0		
5	Produkt 5	1.169	33		
6	Produkt 6	1.329	3		
7	Produkt 7	1.067	6		
8	Produkt 8	1.189	6		
9	Produkt 9	1.397	6		
10	Produkt 10	1.128	6		
11					
12	Anzahl gelbe Zellen		6		
13					

Zählen der Zellen mit einer bestimmten Hintergrundfarbe

Beispiel: Summieren von allen gelben Zellen

Um die gelben Zellen zu summieren, verwenden Sie die **SUMMEWENN-Funktion**.

C13		: ✕ ✓ <i>fx</i>		=SUMMEWENN(C1:C10;6;B1:B10)	
	A	B	C	D	E
1	Produkt 1	1.466	6		
2	Produkt 2	1.652	6		
3	Produkt 3	1.376	44		
4	Produkt 4	1.944	0		
5	Produkt 5	1.169	33		
6	Produkt 6	1.329	3		
7	Produkt 7	1.067	6		
8	Produkt 8	1.189	6		
9	Produkt 9	1.397	6		
10	Produkt 10	1.128	6		
11					
12	Anzahl gelbe Zellen		6		
13	Summe gelbe Zellen		7899		
14					

Addieren der Werte mit einer bestimmten Hintergrundfarbe

Die SUMMEWENN-Funktion in der obigen Abbildung durchsucht die Hilfsspalte C nach dem Farbcode 6 und summiert die entsprechenden Werte in den Zellen in der Spalte B.

Aktualisieren bei Änderung der Hintergrundfarbe

Beachten Sie, dass Änderungen der Hintergrundfarbe nicht automatisch aktualisiert werden. Sie müssen hierzu die **F9-Taste** drücken, um die **Berechnungen zu aktualisieren**.

Datei abspeichern als xlsx-Datei

Da es sich bei der Funktion **ZELLE.ZUORDNEN()** um eine Makrofunktion handelt, müssen Sie die Datei im Format **xlsx** speichern.

Empfehlungen aus dem Management-Handbuch

Werte suchen in Excel

Wie Sie die typische Excel-Aufgabe lösen und aus einer umfangreichen Datentabelle Werte ermitteln, die vorgegebene Kriterien erfüllen. Mit Tipps, Tricks und Beispielen für klassische Excel-Funktionen wie SVERWEIS, INDEX, EINDEUTIG oder ÜBERNEHMEN.

<https://www.business-wissen.de/id/kapitel/316/>

Datenanalyse in Excel

Wie Sie Besonderheiten in Ihren Excel-Daten und Tabellen mit einfachen Formeln erkennen und Ihre Daten so besser bewerten. Mit einfachen Tricks und nützlichen Excel-Funktionen.

<https://www.business-wissen.de/id/kapitel/305/>

Sortieren und Filtern in Excel

In diesem Dokumentenpaket erfahren Sie, wie Sie Ihre Excel-Tabellen schnell und einfach sortieren oder filtern können. Das unterstützt Sie bei Ihrer Arbeit mit Excel-Daten und bei der Datenanalyse. So erstellen Sie im Handumdrehen nützliche Reports.

<https://www.business-wissen.de/id/kapitel/301/>

Nutzen Sie als
Premium-Mitglied
alle
Handbuch-Kapitel
mit mehr als
3.000 Checklisten und Excel-Vorlagen

Jetzt anmelden

www.business-wissen.de/anmelden/

Impressum

b-WISE GmbH Business Wissen Information Service
Bismarckstraße 21
76133 Karlsruhe
DEUTSCHLAND

service@business-wissen.de
Telefon +49 721 18397-0

Copyright 2026, b-wise GmbH, All Rights Reserved